Panel data methods for microeconometrics using Stata

A. Colin Cameron
Univ. of California - Davis

Based on A. Colin Cameron and Pravin K. Trivedi,
Microeconometrics using Stata, Stata Press, forthcoming.

April 8, 2008
1. Introduction

**Panel data** are repeated measures on individuals \((i)\) over time \((t)\).

Regress \(y_{it}\) on \(x_{it}\) for \(i = 1, ..., N\) and \(t = 1, ..., T\).

**Complications** compared to cross-section data:

1. **Inference:** correct (inflate) standard errors. This is because each additional year of data is not independent of previous years.
2. **Modelling:** richer models and estimation methods are possible with repeated measures. Fixed effects and dynamic models are examples.
3. **Methodology:** different areas of applied statistics may apply different methods to the same panel data set.
This talk: overview of panel data methods and xt commands for Stata 10 most commonly used by microeconometricians.

Three specializations to general panel methods:

1. **Short panel:** data on many individual units and few time periods. Then data viewed as clustered on the individual unit. Many panel methods also apply to clustered data such as cross-section individual-level surveys clustered at the village level.

2. **Causation from observational data:** use repeated measures to estimate key marginal effects that are causative rather than mere correlation. Fixed effects: assume time-invariant individual-specific effects. IV: use data from other periods as instruments.

3. **Dynamic models:** regressors include lagged dependent variables.
2.1 Example: wages

- Goal: estimate causative effect of education on wages.
- Complication: education is time-invariant in these data. Rules out fixed effects. Need to use IV methods (Hausman-Taylor).
2.2 Reading in panel data

- Data organization may be
  - long form: each observation is an individual-time \((i, t)\) pair
  - wide form: each observation is data on \(i\) for all time periods
  - wide form: each observation is data on \(t\) for all individuals

- `xt` commands require data in long form
  - use `reshape long` command to convert from wide to long form.

- Data here are already in long form

* Read in data set
  - `use mus08psidextract.dta, clear`
  (PSID wage data 1976-82 from Baltagi and Khanti-Akom (1990))
2.3 Summarize data using usual commands

. * Describe dataset
. describe

Contains data from mus08psidextract.dta

<table>
<thead>
<tr>
<th>Variable</th>
<th>Storage</th>
<th>Display</th>
<th>Value</th>
<th>Label</th>
</tr>
</thead>
<tbody>
<tr>
<td>exp</td>
<td>float</td>
<td>%9.0g</td>
<td></td>
<td>years of full-time work experience</td>
</tr>
<tr>
<td>wks</td>
<td>float</td>
<td>%9.0g</td>
<td></td>
<td>weeks worked</td>
</tr>
<tr>
<td>occ</td>
<td>float</td>
<td>%9.0g</td>
<td></td>
<td>occupation; occ==1 if in a blue-collar</td>
</tr>
<tr>
<td>ind</td>
<td>float</td>
<td>%9.0g</td>
<td></td>
<td>industry; ind==1 if working in a manuf</td>
</tr>
<tr>
<td>south</td>
<td>float</td>
<td>%9.0g</td>
<td></td>
<td>residence; south==1 if in the South area</td>
</tr>
<tr>
<td>smsa</td>
<td>float</td>
<td>%9.0g</td>
<td></td>
<td>smsa==1 if in the Standard metropolis</td>
</tr>
<tr>
<td>ms</td>
<td>float</td>
<td>%9.0g</td>
<td></td>
<td>marital status</td>
</tr>
<tr>
<td>union</td>
<td>float</td>
<td>%9.0g</td>
<td></td>
<td>if wage set be a union contract</td>
</tr>
<tr>
<td>ed</td>
<td>float</td>
<td>%9.0g</td>
<td></td>
<td>years of education</td>
</tr>
<tr>
<td>blk</td>
<td>float</td>
<td>%9.0g</td>
<td></td>
<td>black</td>
</tr>
<tr>
<td>lwage</td>
<td>float</td>
<td>%9.0g</td>
<td></td>
<td>log wage</td>
</tr>
<tr>
<td>id</td>
<td>float</td>
<td>%9.0g</td>
<td></td>
<td></td>
</tr>
<tr>
<td>t</td>
<td>float</td>
<td>%9.0g</td>
<td></td>
<td></td>
</tr>
<tr>
<td>exp2</td>
<td>float</td>
<td>%9.0g</td>
<td></td>
<td></td>
</tr>
</tbody>
</table>
. * Summarize dataset
. summarize

<table>
<thead>
<tr>
<th>Variable</th>
<th>Obs</th>
<th>Mean</th>
<th>Std. Dev.</th>
<th>Min</th>
<th>Max</th>
</tr>
</thead>
<tbody>
<tr>
<td>exp</td>
<td>4165</td>
<td>19.8538</td>
<td>10.96637</td>
<td>1</td>
<td>51</td>
</tr>
<tr>
<td>wks</td>
<td>4165</td>
<td>46.81152</td>
<td>5.129098</td>
<td>5</td>
<td>52</td>
</tr>
<tr>
<td>occ</td>
<td>4165</td>
<td>.5111645</td>
<td>.4999354</td>
<td>0</td>
<td>1</td>
</tr>
<tr>
<td>ind</td>
<td>4165</td>
<td>.3954382</td>
<td>.4890033</td>
<td>0</td>
<td>1</td>
</tr>
<tr>
<td>south</td>
<td>4165</td>
<td>.2902761</td>
<td>.4539442</td>
<td>0</td>
<td>1</td>
</tr>
<tr>
<td>smsa</td>
<td>4165</td>
<td>.6537815</td>
<td>.475821</td>
<td>0</td>
<td>1</td>
</tr>
<tr>
<td>ms</td>
<td>4165</td>
<td>.8144058</td>
<td>.3888256</td>
<td>0</td>
<td>1</td>
</tr>
<tr>
<td>fem</td>
<td>4165</td>
<td>.112605</td>
<td>.3161473</td>
<td>0</td>
<td>1</td>
</tr>
<tr>
<td>union</td>
<td>4165</td>
<td>.3639856</td>
<td>.4812023</td>
<td>0</td>
<td>1</td>
</tr>
<tr>
<td>ed</td>
<td>4165</td>
<td>12.84538</td>
<td>2.787995</td>
<td>4</td>
<td>17</td>
</tr>
<tr>
<td>blk</td>
<td>4165</td>
<td>.0722689</td>
<td>.2589637</td>
<td>0</td>
<td>1</td>
</tr>
<tr>
<td>lwage</td>
<td>4165</td>
<td>6.676346</td>
<td>.4615122</td>
<td>4.60517</td>
<td>8.537</td>
</tr>
<tr>
<td>id</td>
<td>4165</td>
<td>298</td>
<td>171.7821</td>
<td>1</td>
<td>595</td>
</tr>
<tr>
<td>t</td>
<td>4165</td>
<td>4</td>
<td>2.00024</td>
<td>1</td>
<td>7</td>
</tr>
<tr>
<td>exp2</td>
<td>4165</td>
<td>514.405</td>
<td>496.9962</td>
<td>1</td>
<td>2601</td>
</tr>
</tbody>
</table>

Balanced and complete as $7 \times 595 = 4165$. 
. * Organization of data set
. list id t exp wks occ in 1/3, clean

<table>
<thead>
<tr>
<th></th>
<th>1</th>
<th>1</th>
<th>3</th>
<th>32</th>
<th>0</th>
</tr>
</thead>
<tbody>
<tr>
<td>2.</td>
<td>1</td>
<td>2</td>
<td>4</td>
<td>43</td>
<td>0</td>
</tr>
<tr>
<td>3.</td>
<td>1</td>
<td>3</td>
<td>5</td>
<td>40</td>
<td>0</td>
</tr>
</tbody>
</table>

Data are sorted by id and then by t
2.3 Summarize data using xt commands

- xtset command defines $i$ and $t$.
- Allows use of panel commands and some time series operators

```stata
. * Declare individual identifier and time identifier
. xtset id t
panel variable:  id (strongly balanced)
time variable:  t, 1 to 7
delta:        1 unit
```
. * Panel description of data set
. xtdescribe

  id:  1, 2, ..., 595                 n =     595
  t:  1, 2, ..., 7                  T =      7
  Delta(t) = 1 unit
  Span(t) = 7 periods
  (id*t uniquely identifies each observation)

Distribution of T_i:  min    5%     25%     50%     75%     95%     max
                      7      7      7      7      7      7      7

<table>
<thead>
<tr>
<th>Freq.</th>
<th>Percent</th>
<th>Cum.</th>
<th>Pattern</th>
</tr>
</thead>
<tbody>
<tr>
<td>595</td>
<td>100.00</td>
<td>100.00</td>
<td>1111111</td>
</tr>
<tr>
<td>595</td>
<td>100.00</td>
<td></td>
<td>xxxxxxxx</td>
</tr>
</tbody>
</table>

Data are balanced with every individual i having 7 time periods of data.
For time-invariant variable \texttt{ed} the within variation is zero.
For individual-invariant variable \texttt{t} the between variation is zero.
For \texttt{lwage} the within variation \textless between variation.
29.03% on average were in the south.  
**20.59%** were ever in the south.  
94.9% of those ever in the south were always in the south.

```
* Panel tabulation for a variable
xttab south

<table>
<thead>
<tr>
<th>south</th>
<th>Overall Freq.</th>
<th>Overall Percent</th>
<th>Between Freq.</th>
<th>Between Percent</th>
<th>Within Percent</th>
</tr>
</thead>
<tbody>
<tr>
<td>0</td>
<td>2956</td>
<td>70.97%</td>
<td>428</td>
<td>71.93%</td>
<td>98.66%</td>
</tr>
<tr>
<td>1</td>
<td>1209</td>
<td>29.03%</td>
<td>182</td>
<td><strong>30.59%</strong></td>
<td>94.90%</td>
</tr>
<tr>
<td>Total</td>
<td>4165</td>
<td>100.00%</td>
<td>610</td>
<td>102.52%</td>
<td>97.54%</td>
</tr>
</tbody>
</table>

(n = 595)
```
. * Transition probabilities for a variable
. xttrans south, freq

<table>
<thead>
<tr>
<th>residence; south==1 if in the South area</th>
<th>residence; south==1 if in the South area</th>
<th>Total</th>
</tr>
</thead>
<tbody>
<tr>
<td>0</td>
<td>1</td>
<td></td>
</tr>
<tr>
<td>2,527</td>
<td>8</td>
<td>2,535</td>
</tr>
<tr>
<td>99.68</td>
<td>0.32</td>
<td>100.00</td>
</tr>
<tr>
<td>1</td>
<td></td>
<td></td>
</tr>
<tr>
<td>8</td>
<td>1,027</td>
<td>1,035</td>
</tr>
<tr>
<td>0.77</td>
<td>99.23</td>
<td>100.00</td>
</tr>
<tr>
<td>Total</td>
<td></td>
<td></td>
</tr>
<tr>
<td>2,535</td>
<td>1,035</td>
<td>3,570</td>
</tr>
<tr>
<td>71.01</td>
<td>28.99</td>
<td>100.00</td>
</tr>
</tbody>
</table>

For the 28.99% of the sample ever in the south, 99.23% remained in the south the next period.
* Time series plots of log wage for first 10 individuals

```
xtline lwage if id<=10, overlay
```
. * First-order autocorrelation in a variable
. sort id t

. correlate lwage L.lwage L2.lwage L3.lwage L4.lwage L5.lwage L6.lwage
(obs=595)

<table>
<thead>
<tr>
<th></th>
<th>lwage</th>
<th>L. lwage</th>
<th>L2. lwage</th>
<th>L3. lwage</th>
<th>L4. lwage</th>
<th>L5. lwage</th>
<th>L6. lwage</th>
</tr>
</thead>
<tbody>
<tr>
<td>lwage</td>
<td>1.0000</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>L1.</td>
<td>0.9238</td>
<td>1.0000</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>L2.</td>
<td>0.9083</td>
<td>0.9271</td>
<td>1.0000</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>L3.</td>
<td>0.8753</td>
<td>0.8843</td>
<td>0.9067</td>
<td>1.0000</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>L4.</td>
<td>0.8471</td>
<td>0.8551</td>
<td>0.8833</td>
<td>0.8990</td>
<td>1.0000</td>
<td></td>
<td></td>
</tr>
<tr>
<td>L5.</td>
<td>0.8261</td>
<td>0.8347</td>
<td>0.8721</td>
<td>0.8641</td>
<td>0.8667</td>
<td>1.0000</td>
<td></td>
</tr>
<tr>
<td>L6.</td>
<td>0.8033</td>
<td>0.8163</td>
<td>0.8518</td>
<td>0.8465</td>
<td>0.8594</td>
<td>0.9418</td>
<td>1.0000</td>
</tr>
</tbody>
</table>

High serial correlation: Cor[y_t, y_{t-6}] = 0.80.
Can also estimate correlations without imposing stationarity.
Commands *describe, summarize and tabulate* confound cross-section and time series variation.

Instead use **specialized panel commands** after *xtset*:

- **xtdescribe**: extent to which panel is unbalanced
- **xtsum**: separate within (over time) and between (over individuals) variation
- **xttab**: tabulations within and between for discrete data e.g. binary
- **xttrans**: transition frequencies for discrete data
- **xtline**: time series plot for each individual on one chart
- **xtdata**: scatterplots for within and between variation.
2.4 Pooled OLS

Do regular OLS of $y_{it}$ on $x_{it}$.

```
. * Pooled OLS with incorrect default standard errors
. regress lwage exp exp2 wks ed, noheader

| lwage | Coef.  | Std. Err. | t     | P>|t| | [95% Conf. Interval] |
|-------|--------|-----------|-------|------|---------------------|
| exp   | 0.044675 | 0.0023929 | 18.67 | 0.000 | 0.0399838 - 0.0493663 |
| exp2  | -0.0007156 | 0.0000528 | -13.56 | 0.000 | -0.0008191 - 0.0006121 |
| wks   | 0.005827 | 0.0011827 | 4.93  | 0.000 | 0.0035084 - 0.0081456 |
| ed    | 0.0760407 | 0.0022266 | 34.15 | 0.000 | 0.0716754 - 0.080406 |
| _cons | 4.907961 | 0.0673297 | 72.89 | 0.000 | 4.775959  - 5.039963 |
```

The default standard errors erroneously assume errors are independent over $i$ for given $t$.
Assumes more information content from data then is the case.
. * Pooled OLS with cluster-robust standard errors
. regress lwage exp exp2 wks ed, noheader vce(cluster id)

                 Coef.     Std. Err.      t    P>|t|     [95% Conf. Interval]
      lwage
                exp   0.044675   0.0054385    8.21   0.000     0.0339941   0.055356
                exp2  -0.0007156   0.0001285   -5.57   0.000   -0.0009679   -0.0004633
                wks    0.005827   0.0019284     3.02   0.003      0.0020396    0.0096144
                ed    0.0760407   0.0052122    14.59   0.000     0.0658042    0.0862772
               _cons   4.907961   0.1399887    35.06   0.000     4.633028    5.182894

Cluster-robust standard errors are twice as large as default.
Cluster-robust t-statistics are half as large as default.
Typical result. Need to use cluster-robust se's if use pooled OLS.
3.1 Some basic considerations

1. **Regular time intervals** assumed.
2. **Unbalanced** panel okay (xt commands handle unbalanced data).
   [Should then rule out selection/attrition bias].
3. **Short panel** assumed, with $T$ small and $N \to \infty$.
   [Versus long panels, with $T \to \infty$ and $N$ small or $N \to \infty$.]
4. **Errors are correlated.**
   [For short panel: correlated over $t$ for given $i$, but not over $i$.]
5. **Parameters** may vary over individuals or time.
   **Intercept**: Individual-specific effects model (fixed or random effects).
   **Slopes**: Pooling and random coefficients models.
6. **Regressors**: time-invariant, individual-invariant, or vary over both.
7. **Prediction**: ignored.
   [Not always possible even if marginal effects computed.]
8. **Dynamic models**: possible.
   [Usually static models are estimated.]
3.2 Basic linear panel models

- **Pooled model (or population-averaged)**
  \[ y_{it} = \alpha + x'_{it} \beta + u_{it}. \] (1)

- **Two-way effects model** allows intercept to vary over \( i \) and \( t \)
  \[ y_{it} = \alpha_i + \gamma_t + x'_{it} \beta + \varepsilon_{it}. \] (2)

- **Individual-specific effects model**
  \[ y_{it} = \alpha_i + x'_{it} \beta + \varepsilon_{it}, \] (3)
  where \( \alpha_i \) may be fixed effect or random effect.

- **Mixed model or random coefficients model** allows slopes to vary over \( i \)
  \[ y_{it} = \alpha_i + x'_{it} \beta_i + \varepsilon_{it}. \] (4)
3.3 Fixed effects versus random effects

- **Individual-specific effects model:**
  \[ y_{it} = x_{it}' \beta + (\alpha_i + \varepsilon_{it}). \]

- **Fixed effects (FE):**
  - \( \alpha_i \) is a random variable possibly correlated with \( x_{it} \)
  - so regressor \( x_{it} \) may be **endogenous** (wrt to \( \alpha_i \) but not \( \varepsilon_{it} \))
  - e.g. education is correlated with time-invariant ability
  - pooled OLS, pooled GLS, RE are inconsistent for \( \beta \)
  - within (FE) and first difference estimators are consistent.

- **Random effects (RE) or population-averaged (PA):**
  - \( \alpha_i \) is purely random (usually iid \( (0, \sigma^2_{\alpha}) \)) unrelated to \( x_{it} \)
  - so regressor \( x_{it} \) is exogenous
  - all estimators are consistent for \( \beta \)

- **Fundamental divide:** microeconometricians FE versus others RE.
3.4 Cluster-robust inference

- Many methods assume $\varepsilon_{it}$ and $\alpha_i$ (if present) are iid.
- Yields **wrong standard errors** if heteroskedasticity or if errors not equicorrelated over time for a given individual.
- For short panel can relax and use **cluster-robust inference**.
  - Allows heteroskedasticity and general correlation over time for given $i$.
  - Independence over $i$ is still assumed.
- For `xtreg` use option `vce(robust)` does cluster-robust
- For some other `xt` commands use option `vce(cluster)`
- And for some other `xt` commands there is no option but may be able to do a cluster bootstrap.
4.1 Pooled GLS estimator: *xtgee*

- Regress $y_{it}$ on $x_{it}$ using feasible GLS as error is not iid.

```
. * Pooled FGLS estimator with AR(2) error &
  cluster-robust se’s
. xtgee lwage exp exp2 wks ed, corr(ar 2) vce(robust)
```
4.2 Between GLS estimator: `xtreg, be`

- OLS of $\bar{y}_i$ on $\bar{x}_i$. i.e. Regression using each individual’s averages.

```
. * Between estimator with default standard errors
. xtreg lwage exp exp2 wks ed, be
```
4.3 Random effects estimator: `xtreg, re`

- FGLS in RE model assuming $\alpha_i$ iid $(0, \sigma^2_\alpha)$ and $\varepsilon_i$ iid $(0, \sigma^2_\varepsilon)$.

- Equals OLS of $(y_{it} - \hat{\theta}_i \bar{y}_i)$ on $(x_{it} - \hat{\theta}_i \bar{x}_i)$;
  \[
  \theta_i = 1 - \frac{\sigma^2_\varepsilon}{(T_i \sigma^2_\alpha + \sigma^2_\varepsilon)}.
  \]

* Random effects estimator with cluster-robust se’s

```
. xtreg lwage exp exp2 wks ed, re vce(robust) theta
```

This gives $\hat{\theta} = 0.82$. 
4.4 Fixed effects (or within) estimator: `xtreg, fe`

- OLS regress \((y_{it} - \bar{y}_i)\) on \((x_{it} - \bar{x}_i)\).
- Mean-differencing eliminates \(\alpha_i\) in \(y_{it} = \alpha_i + x'_{it}\beta + \epsilon_{it}\).

. * Within or FE estimator with cluster-robust se’s
. `xtreg lwage exp exp2 wks ed, fe vce(robust)`
4.5 First differences estimator: regress with differences

- OLS regress \((y_{it} - y_{i,t-1})\) on \((x_{it} - x_{i,t-1})\).
- First-differencing eliminates \(\alpha_i\) in \(y_{it} = \alpha_i + x'_{it}\beta + \varepsilon_{it}\).

* First difference estimator with cluster-robust se’s
  . regress D.(lwage $xlist), vce(cluster id)
4.6 Estimator comparison

* Compare various estimators (with cluster-robust se’s)

. global xlist exp exp2 wks ed
. quietly regress lwage $xlist, vce(cluster id)
. estimates store OLS
. quietly xtgee lwage exp exp2 wks ed, corr(ar 2) vce(robust)
. estimates store PFGLS
. quietly xtreg lwage $xlist, be
. estimates store BE
. quietly xtreg lwage $xlist, re vce(robust)
. estimates store RE
. quietly xtreg lwage $xlist, fe vce(robust)
. estimates store FE
. estimates table OLS PFGLS BE RE FE, b(%9.4f) se stats(N)
Coefficients vary considerably across OLS, RE, FE and RE estimators.

FE and RE similar as $\hat{\theta} = 0.82 \approx 1$.

Not shown is that even for FE and RE cluster-robust changes se’s.

Coefficient of ed not identified for FE as time-invariant regressor!

<table>
<thead>
<tr>
<th>Variable</th>
<th>OLS</th>
<th>PFGLS</th>
<th>BE</th>
<th>RE</th>
<th>FE</th>
</tr>
</thead>
<tbody>
<tr>
<td>exp</td>
<td>0.0447</td>
<td>0.0719</td>
<td>0.0382</td>
<td>0.0889</td>
<td>0.1138</td>
</tr>
<tr>
<td>exp2</td>
<td>-0.0007</td>
<td>-0.0009</td>
<td>-0.0006</td>
<td>-0.0008</td>
<td>-0.0004</td>
</tr>
<tr>
<td>wks</td>
<td>0.0001</td>
<td>0.0001</td>
<td>0.0001</td>
<td>0.0001</td>
<td>0.0001</td>
</tr>
<tr>
<td>ed</td>
<td>0.0760</td>
<td>0.0905</td>
<td>0.0738</td>
<td>0.1117</td>
<td>0.0000</td>
</tr>
<tr>
<td>_cons</td>
<td>4.9080</td>
<td>4.5264</td>
<td>4.6830</td>
<td>3.8294</td>
<td>4.5964</td>
</tr>
</tbody>
</table>

Legend: b/se
4.7 Fixed effects versus random effects

- Prefer RE as can estimate all parameters and more efficient.
- But RE is inconsistent if fixed effects present.
- Use **Hausman test** to discriminate between FE and RE.
  - This tests difference between FE and RE estimates is statistically significantly different from zero.
- Problem: **hausman** command gives incorrect statistic as it assumes RE estimator is fully efficient, usually not the case.
- Solution: do a panel bootstrap of the Hausman test or use the Wooldridge (2002) robust version of Hausman test.
### 4.8 Stata linear panel commands

**Panel summary**  
- xtset; xtdescribe; xtsum; xtdata; xtline; xttab; xttran

**Pooled OLS**  
- regress

**Feasible GLS**  
- xtggee, family(gaussian)
- xtgls; xtpcse

**Random effects**  
- xtreg, re; xtregar, re

**Fixed effects**  
- xtreg, fe; xtregar, fe

**Random slopes**  
- xtmixed; quadchk; xtrc

**First differences**  
- regress (with differenced data)

**Static IV**  
- xtivreg; xthtaylor

**Dynamic IV**  
- xtabond; xtdpdsys; xtdpd
5.1 Long panels

- For **short panels** asymptotics are $T$ fixed and $N \to \infty$.
- For **long panels** asymptotics are for $T \to \infty$
  - A dynamic model for the errors is specified, such as AR(1) error
  - Errors may be correlated over individuals
  - Individual-specific effects can be just individual dummies
  - Furthermore if $N$ is small and $T$ large can allow slopes to differ across individuals and test for poolability.
5.2 Commands for long panels

- Models with **stationary errors**:  
  - `xtgls` allows several different models for the error  
  - `xtpcse` is a variation of `xtgls`  
  - `xtregar` does FE and RE with AR(1) error  
  - Add-on `xtscc` gives HAC se’s with spatial correlation.

- Models with **nonstationary errors** (currently active area):  
  - As yet no Stata commands  
  - Add-on `levinlin` does Levin-Lin-Chu (2002) panel unit root test  
  - Add-on `ipshin` does Im-Pesaran-Shin (1997) panel unit root test in heterogeneous panels  
  - Add-on `xtpmg` for does Pesaran-Smith and Pesaran-Shin-Smith estimation for nonstationary heterogeneous panels with both $N$ and $T$ large.
Command `xtivreg` is a natural extension of `ivregress` to panels.

Consider a model with possibly transformed variables:

\[ y_{it}^* = \alpha + x_{it}^* \beta + u_{it}, \]

where transformations are:

- OLS: \( y_{it}^* = y_{it} \)
- Between: \( y_{it}^* = \bar{y}_i \)
- Fixed effects: \( y_{it}^* = (y_{it} - \bar{y}_i) \)
- Random effects: \( y_{it}^* = (y_{it} - \theta_i \bar{y}_i) \)

- **OLS is inconsistent** if \( E[u_{it} | x_{it}^*] = 0 \).

- **IV estimation with instruments** \( z_{it}^* \) satisfy \( E[u_{it} | z_{it}^*] = 0 \).

- Example: `xtivreg lwage exp exp2 (wks = ms), fe`
6.2 Hausman-Taylor IV estimator: xthtaylor

- Command `xthtaylor` uses exogenous time-varying regressors $x_{it}$ from periods other than the current as instruments.
- This enables estimation of coefficient of a time-invariant regressor in a fixed effects model (not possible using FE estimator).
- Example: allows estimation of coefficient of time-invariant regressor ed

```
xthtaylor lwage occ south smsa ind exp exp2 wks ms union fem blk ed, ///
endog(exp exp2 wks ms union ed)
```
Simple dynamic model regresses $y_{it}$ in **polynomial in time**.
- e.g. Growth curve of child height or IQ as grow older
- use previous models with $x_{it}$ polynomial in time or age.

Richer dynamic model regresses $y_{it}$ on **lags** of $y_{it}$.
Leading example: AR(1) model with individual specific effects

\[ y_{it} = \gamma y_{i,t-1} + x_{it}' \beta + \alpha_i + \varepsilon_{it}. \]

Four reasons for \( y_{it} \) being serially correlated over time:

- **True state dependence:** via \( y_{i,t-1} \)
- **Observed heterogeneity:** via \( x_{it} \) which may be serially correlated
- **Unobserved heterogeneity:** via \( \alpha_i \)
- **Error correlation:** via \( \varepsilon_{it} \)

Focus on case where \( \alpha_i \) is a **fixed effect**

- FE estimator is now inconsistent (if short panel)
- Instead use Arellano-Bond estimator
7.3 Arellano-Bond estimator

- **First-difference** to eliminate \( \alpha_i \) (rather than mean-difference)

\[
(y_{it} - y_{i,t-1}) = \gamma(y_{i,t-1} - y_{i,t-2}) + (x_{it} - x'_{i,t-1})\beta + (\varepsilon_{it} - \varepsilon_{i,t-1}).
\]

- **OLS inconsistent** as \((y_{i,t-1} - y_{i,t-2})\) correlated with \((\varepsilon_{it} - \varepsilon_{i,t-1})\)
  (even under assumption \(\varepsilon_{it}\) is serially uncorrelated).

- But \(y_{i,t-2}\) is not correlated with \((\varepsilon_{it} - \varepsilon_{i,t-1})\), so can use \(y_{i,t-2}\) as an instrument for \((y_{i,t-1} - y_{i,t-2})\).

- Arellano-Bond is a variation that uses **unbalanced set** of instruments with further lags as instruments.
  For \(t = 3\) can use \(y_{i1}\), for \(t = 4\) can use \(y_{i1}\) and \(y_{i2}\), and so on.

- Stata commands
  - `xtabond` for Arellano-Bond
  - `xtdpdsys` for Blundell-Bond (more efficient than `xtabond`)
  - `xtdpd` for more complicated models than `xtabond` and `xtdpdsys`.
* Optimal or two-step GMM for a dynamic panel model
.xtabond lwage occ south smsa ind, lags(2) maxldep(3) ///
.pre(wks,lag(1,2)) endogenous(ms,lag(0,2)) ///
.endogenous(union,lag(0,2)) twostep vce(robust) artests(3)

* Test whether error is serially correlated
.estat abond

* Test of overidentifying restrictions
.estat sargan

* Arellano/Bover or Blundell/Bond for a dynamic panel model
.xtdpdsys lwage occ south smsa ind, lags(2) maxldep(3) ///
.pre(wks,lag(1,2)) endogenous(ms,lag(0,2)) ///
.endogenous(union,lag(0,2)) twostep vce(robust) artests(3)
8.1 Random coefficients model: xtrc or xtmixed

- Generalize random effects model to random slopes.
- Command `xtrc` estimates the random coefficients model

\[ y_{it} = \alpha_i + x'_{it}\beta_i + \epsilon_{it}, \]

where \((\alpha_i, \beta_i)\) are iid with mean \((\alpha, \beta)\) and variance matrix \(\Sigma\) and \(\epsilon_{it}\) is iid.
8.2 Mixed or multi-level or hierarchical model: `xtmixed`

- Not used in microeconometrics but used in many other disciplines.
- Stack all observations for individual $i$ and specify

$$y_i = X_i \beta + Z_i u_i + \varepsilon_i$$

where $u_i$ is iid $(0, G)$ and $Z_i$ is called a design matrix.

- Random effects: $Z_i = e$ (a vector of ones) and $u_i = \alpha_i$
- Random coefficients: $Z_i = X_i$.
- Example:

`xtmixed lwage exp exp2 wks ed || id: exp wks, covar(unstructured) mle`
9.1 Clustered data

- Consider data on individual $i$ in village $j$ with clustering on village.
- A cluster-specific model (here village-specific) specifies
  \[ y_{ji} = \alpha_i + x'_{ji} \beta + \epsilon_{ji}. \]
- Here clustering is on village (not individual) and the repeated measures are over individuals (not time).
- Use `xtset village id`
- Assuming equicorrelated errors can be more reasonable here than with panel data (where correlation dampens over time).
  So perhaps less need for `vce(cluster)` after `xtreg`
First use `xtset village person` (versus `xtset id t` for panel).

If $\alpha_i$ is **random** use:

- `regress` with option `vce(cluster village)`
- `xtreg, re`
- `xtgee` with option `exchangeable`
- `xtmixed` for richer models of error structure

If $\alpha_i$ is **fixed** use:

- `xtreg, fe`
10.1 Nonlinear panel models overview

- **General approaches** similar to linear case
  - Pooled estimation or population-averaged
  - Random effects
  - Fixed effects

- **Complications**
  - Random effects often not tractable so need numerical integration
  - Fixed effects models in short panels are generally not estimable due to the incidental parameters problem.

- Here we consider **short panels** throughout.

- **Standard nonlinear models** are:
  - Binary: logit and probit
  - Counts: Poisson and negative binomial
  - Truncated: Tobit
10.2 Nonlinear panel models

- A **pooled** or **population-averaged model** may be used. This is same model as in cross-section case, with adjustment for correlation over time for a given individual.

- A **fully parametric model** may be specified, with conditional density

\[
f(y_{it} | \alpha_i, x_{it}) = f(y_{it}, \alpha_i + x_{it}' \beta, \gamma), \quad t = 1, ..., T_i, \ i = 1, ..., N, \tag{5}\]

where \( \gamma \) denotes additional model parameters such as variance parameters and \( \alpha_i \) is an individual effect.

- A **conditional mean model** may be specified, with **additive effects**

\[
E[y_{it} | \alpha_i, x_{it}] = \alpha_i + g(x_{it}' \beta) \tag{6}
\]

or **multiplicative effects**

\[
E[y_{it} | \alpha_i, x_{it}] = \alpha_i \times g(x_{it}' \beta). \tag{7}\]
10.3 Nonlinear panel commands

<table>
<thead>
<tr>
<th></th>
<th>Counts</th>
<th>Binary</th>
</tr>
</thead>
<tbody>
<tr>
<td><strong>Pooled</strong></td>
<td>poisson</td>
<td>logit</td>
</tr>
<tr>
<td></td>
<td>nbreg</td>
<td>probit</td>
</tr>
<tr>
<td><strong>GEE (PA)</strong></td>
<td><code>xtgee,family(poisson)</code></td>
<td><code>xtgee,family(binomial)</code></td>
</tr>
<tr>
<td></td>
<td><code>xtgee,family(nbinomial)</code></td>
<td><code>link(logit)</code></td>
</tr>
<tr>
<td><strong>RE</strong></td>
<td><code>xtpoisson, re</code></td>
<td><code>xtlogit, re</code></td>
</tr>
<tr>
<td></td>
<td><code>xtnbreg, fe</code></td>
<td><code>xtprobit, re</code></td>
</tr>
<tr>
<td><strong>Random slopes</strong></td>
<td><code>xtmepoisson</code></td>
<td><code>xtmelogit</code></td>
</tr>
<tr>
<td><strong>FE</strong></td>
<td><code>xtpoisson, fe</code></td>
<td><code>xtlogit, fe</code></td>
</tr>
<tr>
<td></td>
<td><code>xtnbreg, fe</code></td>
<td></td>
</tr>
</tbody>
</table>

plus tobit and `xttobit`.

Panel methods for Stata
April 8, 2008
11.1 Pooled or Population-averaged estimation

- Extend pooled OLS
  - Give the usual cross-section command for conditional mean models or conditional density models but then get cluster-robust standard errors
  - Probit example:
    probit y x, vce(cluster id)
    or
    xtgee y x, fam(binomial) link(probit) corr(ind) vce(cluster id)

- Extend pooled feasible GLS
  - Estimate with an assumed correlation structure over time
  - Equicorrelated probit example:
    xtprobit y x, pa vce(boot)
    or
    xtgee y x, fam(binomial) link(probit) corr(exch) vce(cluster id)
11.2 Random effects estimation

- Assume individual-specific effect $\alpha_i$ has specified distribution $g(\alpha_i | \eta)$.
- Then the unconditional density for the $i^{th}$ observation is

$$f(y_{i1}, ..., y_{iT} | x_{i1}, ..., x_{iT}, \beta, \gamma, \eta) = \int \left[ \prod_{t=1}^{T} f(y_{it} | x_{it}, \alpha_i, \beta, \gamma) \right] g(\alpha_i | \eta) d\alpha_i. \quad (8)$$

- **Analytical solution:**
  - For Poisson with gamma random effect
  - For negative binomial with gamma effect
  - Use `xtpoisson, re` and `xtnbreg, re`

- **No analytical solution:**
  - For other models.
  - Instead use numerical integration (only univariate integration is required).
  - Assume normally distributed random effects.
  - Use `re` option for `xtlogit`, `xtprobit`
  - Use `normal` option for `xtpoisson` and `xtnbreg`
11.2 Random slopes estimation

- Can extend to **random slopes**.
  - Nonlinear generalization of `xtmixed`
  - Then higher-dimensional numerical integral.
  - Use adaptive Gaussian quadrature

- **Stata commands are:**
  - `xtmelogit` for binary data
  - `xtmepoisson` for counts

- **Stata add-on that is very rich:**
  - `gllamm` (generalized linear and latent mixed models)
  - Developed by Sophia Rabe-Hesketh and Anders Skrondal.
In general not possible in short panels.

**Incidental parameters problem:**
- $N$ fixed effects $\alpha_i$ plus $K$ regressors means $(N + K)$ parameters
- But $(N + K) \to \infty$ as $N \to \infty$
- Need to eliminate $\alpha_i$ by some sort of differencing
  - possible for Poisson, negative binomial and logit.

**Stata commands**
- `xtlogit, fe`
- `xtpoisson, fe` (better to use `xtpqml` as robust se’s)
- `xtnbreg, fe`

Fixed effects extended to **dynamic models** for logit and probit. No Stata command.
Stata provides commands for panel models and estimators commonly used in microeconometrics and biostatistics.

Stata also provides diagnostics and postestimation commands, not presented here.

The emphasis is on short panels. Some commands provide cluster-robust standard errors, some do not.

A big distinction is between fixed effects models, emphasized by microeconometricians, and random effects and mixed models favored by many others.

Extensions to nonlinear panel models exist, though FE models may not be estimable with short panels.

This presentation draws on two chapters in Cameron and Trivedi, *Microeconometrics using Stata*, forthcoming.
For Cameron and Trivedi, *Microeconometrics using Stata*, forthcoming.

1. Stata basics
2. Data management and graphics
3. Linear regression basics
4. Simulation
5. GLS regression
6. Linear instrumental variable regression
7. Quantile regression
8. Linear panel models
9. Nonlinear regression methods
10. Nonlinear optimization methods
11. Testing methods
12. Bootstrap methods
13. Binary outcome models
14. Multinomial models
15. Tobit and selection models
16. Count models
17. Nonlinear panel models
18. Topics
   A. Programming in Stata
   B. Mata
Econometrics graduate-level panel data texts

- Comprehensive panel texts

- More selective advanced panel texts

- Texts with several chapters on panel